Using Sax Basic Engine with Visual Basic

In order to use Sax Basic Engine with Visual Basic, you need to follow several steps:

1) Create an ActiveX.EXE project.

2) Modify the project properties to work with Sax Basic Engine.

3) Add the necessary components to your project.

4) Add proper shutdown and DoEvents routines.

In the following sections, we’ll step through the creation of a simple example demonstrating how to use Sax Basic Engine with your project.  Our example will create a simple ‘Hello World’ program.

Creating an ActiveX.EXE Project

Sax Basic Engine uses COM Automation to extend its language.  In order to do this, you need to create an ActiveX EXE project instead of the standard EXE project, which Visual Basic defaults to.  Because Visual Basic does not support OLE automation from a standard executable, Sax Basic Engine will not be able to communicate with your program unless an ActiveX EXE project is created.

Modifying the Project Properties

Visual Basic uses default settings for an ActiveX EXE project that must be changed for your program to work.  You can access the project properties by selecting the Project | <project> Properties menu choice.  

· On the General tab, change the Startup Object to Sub Main.

· On the Component tab, change the Start Mode to Standalone.

Figure 1 - Changing Project Properties
Adding the Necessary Components to your Project

Now that your project is configured correctly, you need to add a few components.  When you create an ActiveX EXE project it only creates a single multi-use class.  For Sax Basic Engine to use a class, it must be set to this type of instancing since this provides the interface for automation.

To get started, you will need a form for your Sax Basic Engine control to reside on.  Create a simple new form and name it frmBasic (for example).  Next, select the Project | Components menu item and add Sax Basic Engine – Enterprise Library v6 to your project.  You should now have a control in your control toolbar that represents Sax Basic Engine.  Add this control to your form. You will see two controls, one called BasicIdeCtl and one called BasicNoUICtl.  The second control is Sax Basic Engine without the IDE, so for demonstration purposes here, we will use the first control.

Because you set your project to use Sub Main to start, you will need to create a new module that has your Sub Main definition in it such as that shown below.

Sub Main()

    frmBasic.Show

End Sub

Adding Proper Shutdown and DoEvents Routines

To shut down Sax Basic Engine properly, you will need to add a timer and some code to your form.  During your QueryUnload event, it is necessary to request a shutdown of the Sax Basic Engine using the Shutdown method.  If this method returns 0, it is okay to close your form.  If it returns -1, you should set your timer to a small value and attempt to unload again after this time has expired.  If it returns 1, then the shutdown was cancelled by the user and you may cancel the unload request.  The following code demonstrates this.

Private Sub Form_QueryUnload(Cancel As Integer, UnloadMode As Integer)

    Select Case BasicIdeCtl1.Shutdown

        Case -1 ' basic engine execution is nested

            Timer1.Interval = 100

            Cancel = True

        Case 0 ' Shutdown Okay

        Case 1 ' Shutdown canceled by user

            Cancel = True

    End Select

End Sub

Private Sub Form_Unload(Cancel As Integer)

    Cancel = Not BasicIdeCtl1.Disconnect

End Sub

Private Sub Timer_Timer()

    Timer.Interval = 0

    If Not BasicIdeCtl1.Run Then Unload Me

End Sub

It is also necessary to add the following code to your DoEvents event.

Private Sub BasicIdeCtl1_DoEvents()

    DoEvents

End Sub

This will ensure that your program will still respond when Sax Basic Engine is busy executing code.

Extending the Basic Language

There are two basic things you will probably want to do with Sax Basic Engine.  First, you will want to be able to write code in Sax Basic Engine which calls code in your program or other automation objects.  Second, you will want to be able to call code written in Sax Basic Engine from your program (which is covered in the next chapter).  The first one is called ‘extending’ the Sax Basic Engine language.  We extend the language by using Automation objects, whether it be by automation classes we have written or through type libraries already defined.

Creating an Automation Class for Sax Basic Engine

When you create an ActiveX EXE project an automation class is created by default.  We will use this class to provide automation for our programs, but you can add your own automation classes (with instancing set to multi-use) or you can use the type libraries of any ActiveX control to extend the language.

For the class, enter in the following code:

Public Sub HelloWorld()

    MsgBox "Hello World!"

End Sub

For this example, save your class with the name “CHello”.  This creates a simple subroutine for our class that Sax Basic Engine will be able to call.

Adding Extensions to Sax Basic Engine from a Class

Next, we will add the extension during the loading of the Sax Basic Engine form.

Private Sub Form_Load()

    If Not BasicIdeCtl1.AddExtension("", New CHello) Then

        MsgBox "Error Adding CHello class."

    End If

End Sub

Using the default method for AddExtension, we will give the user the ability to call any public sub, function, or property from our class as if it were part of the language.  In this case, the user will be able to call HelloWorld in any part of their macro without the need to declare an instance of our Chello class.  There are many ways you can add extensions, which are covered later in this chapter.

Testing Hello World

Now, run your program.  When your program begins it will add the automation class as an extension to the Sax Basic Engine Language.  It will start with a blank sheet that includes a Sub Main() routine in it.  Inside this routine, add the following line of code:

Sub Main()

  HelloWorld

End Sub

Now, click on the run button in the Sax Basic Engine IDE and you should have a screen similar to that shown below.

Figure 2 - Running 'Hello World'

Congratulations, you have just run your first Sax Basic Engine program!

Variations on Adding Extensions

There are several different ways add extensions to the Sax Basic language.  The table below summarizes the different ways you can use AddExtension.

Scan Character
Object Added
Object Type Added
Global constants in type library added
All types referenced directly or indirectly by Object added
All Objects in containing type library added

None
Yes
No
Yes
Yes
No

‘+’
Yes
Yes
Yes
No
Yes

‘-‘
Yes
No
No
No
No

‘*’
No
No
Yes
Yes
No

‘~’
No
Yes
No
No
No

‘=’
Yes
Yes
Yes
Yes
No

‘.’
Yes
Yes
No
No
No

For example, if we only wanted to add only the hello class to our Hello World example, we could use the following code:

If Not BasicIdeCtl1.AddExtension(‘~’, hello) Then


MsgBox "Unable to add the hello type to Sax Basic Engine"

End If

Extending Sax Basic with ActiveX controls and Type Libraries

Just as we can create our own automation classes and extend the Sax Basic language with them, we can also extend it with any ActiveX control or Type Library.  In our next example, we will use AddExtension with an ActiveX control to Sax Basic Engine.  We begin by adding our base code to our ActiveX EXE program as shown in the previous section.  Next, we’ll draw our Sax Basic Engine control on our form, leaving room for a text box control.

After this, we’ll add the Microsoft Rich TextBox Control to our form and the following code to add it as an extension to Sax Basic Engine:

Private Sub Form_Load()


BasicIdeCtl1.AddExtension "-rtf.", rtfTextBox

End Sub

Using AddExtension in this manner will only add the object to Sax Basic Engine and require ‘rtf.’ to be prefaced on all object properties and methods (see the following figure).

Figure 3 - Adding ActiveX Controls as Extensions

Once we click on the run button, Sax Basic Engine will begin execution at Sub Main(), call rtf.BackColor to change the background color of our Rich Text Box, and then set the .Text property to “Hello World!”.

Executing Code Inside Sax Basic Engine Modules

In the previous examples, we have shown how to extend the basic language with classes and ActiveX controls which allow you to execute code that is internal to your program from Sax Basic Engine.  In this section we will discuss the different methods for calling code that exists inside of Sax Basic Engine.

The Run Property

The most basic way to run a module is to have it loaded in the IDE and then click on the run button.  Internally this sets the Run property to TRUE and begins execution of the module at its Sub Main().  We can set the property at run time through our code to execute any module we want to.  For example:

Public Sub ExecuteCurrentModule()

  BasicIdeCtl1.Run = TRUE

End Sub

However, this is somewhat limited in that regular (non-event driven) mode we will only execute the current module in the IDE.

The Code Property

Sax Basic Engine allows us to specify or modify the code currently in the IDE.  We can do this by referencing the Code property.  By changing the Code property, we can change the current macro and execute it by setting the Run property to TRUE.

The RunThis Method

We can also call any macro code by using the RunThis Method.  RunThis allows us to pass the name of any method or property along with its parameters to Sax Basic Engine.  In our Hello World example, we could have created used the listing below to call our HelloWorld sub.

BasicIdeCtl1.RunThis "HelloWorld"

You can also pass parameters to any function or sub as demonstrated below.

BasicIdeCtl1.RunThis "MsgBox ""Hello World!"""

Note that with RunThis you cannot get a return value from it.  In order to get return values from your macro calls you need to use the Evaluate property, the Module object, or an Event handler object.

The Evaluate Property

The Evaluate property can be used to get and set values for expressions in Sax Basic Engine.  By passing Sax Basic Engine a string it will execute the string in an attempt to evaluate it.  There  are two way we can use the Evaluate property.  First, we can use it to return values:

result = BasicIdeCtl1.Evaluate("x + 1")

If x is a variable already defined somewhere in a module, this will return a string which is x incremented by 1.  Note that we can use any expression in our string as long as it translates into useable Basic code.  If we had a function called Add that allowed you to pass two variables to it, we could write the following code:

x = 1

y = 1

result = BasicIdeCtl1.Evaluate("Add(" & x & ", " & y & ")")

The variable result would be equal to two in this case.

The second way to use the Evaluate property is by assigning it a value.  If we have a variable called x defined in a macro, we could call Evaluate and assign it a value as follows:

BasicIdeCtl1.Evaluate("x") = 1

In this case, the value of x is now 1.

The RunFile Method

We can also execute a macro file directly from file or memory by specifying it with the RunFile method:

BasicIdeCtl1.RunFile "MyModule.Bas"

This allows us to load and execute the module immediately (while suspending the currently executing module).  Note that it will begin execution at your Sub Main().

The Module Object

A more convenient way to execute code in modules is to create a module object, which contains all public properties and methods of the module loaded.  In our next example, we will do this to create a simple calculator.  First we create our standard Basic Engine project through the methods outlined previously.  Next we’ll add our main form with two text boxes for input, a label for output, and a button to execute our macro.

Figure 4 - Calculator Main Form
For our macro code, we create a create a file called “calc.bas” with the following code:

Public Function Calculate(x As Long, y As Long) As Long


Calculate = x - y

End Function

When a user clicks on the Calculate button, we’ll call the Calculate function from our macro file through Sax Basic Engine like this:

lblResult.Caption = Formula.Calculate(Val(txtX), Val(txtY))

Notice that Calculate is called like we would any other method of an object.  Formula is our Module object that is created during the loading of our main form with the following code:

Set Formula = frmBasic.BasicIdeCtl1.ModuleInstance("calc.bas", False)

There is one last thing we need to do here.  Because you can’t edit a module which is currently loaded, we need to delete this instance in order give the user the ability to edit it.  Using the code listing below, we set the instance to nothing, load the module in the IDE, and then recreate the instance.  If the user has made changes and saved them, we will pick up the new changes and use them.

Set Formula = Nothing

frmBasic.BasicIdeCtl1.filename = "calc.bas"

frmBasic.Show vbModal, Me

Set Formula = frmBasic.BasicIdeCtl1.ModuleInstance("calc.bas", False)

Now if we run our program we have the ability to change our calculation function whenever we wish.

The Handler Object

The Handler Object allows us to create events which can then be defined in Sax Basic Engine.  In order to create event handlers, we must first declare them, then place Sax Basic Engine in event mode, and then call our events.

Taking the previous example of the calculator, we will replace our module object with an event handler.  Our load event on our main form is defined as follows:

frmBasic.BasicIdeCtl1.EventMode = True

Set Formula = frmBasic.BasicIdeCtl1.CreateHandler _

        ("Function Calculate(x As Long, y As Long) As Long")

frmBasic.BasicIdeCtl1.LoadModule "calc.bas"

frmBasic.BasicIdeCtl1.Run = True

With a handler, we are pointing to only one method or property so we don’t have to specify the name when calling it, so we will redefine our Click event of our Calculate button as follows:

lblResult.Caption = Formula.Evaluate(Val(txtX), Val(txtY))

Finally, we have to change how we edit this macro from Sax Basic Engine.  First we need to unload the module so we can edit it.  When we’re done editing, we can reload the module.

frmBasic.BasicIdeCtl1.filename = "calc.bas"

frmBasic.BasicIdeCtl1.UnloadModule "calc.bas"

Set Formula = Nothing

frmBasic.Show vbModal, Me

frmBasic.BasicIdeCtl1.LoadModule "calc.bas"

With a handler, we are pointing to only one method so we don’t have to specify the name when calling it.  Now when we run our program we can execute the macros by defining them as events and creating handlers.

Source code

The source code covered can be found in the same directory as this documentation.

